Step-by-step use of Zeilberger's algorithm

```
> restart;
> a:=binomial(n,2*k)*binomial(2*k,k)/4^k;
```

$$a := \frac{\binom{n}{2k}\binom{2k}{k}}{4^k} \quad (1)$$

Order 1

We start by looking for a recurrence of order 1.

```
> N:=1:
> telescoper:=add(lambda[N-i](n)*u(n+i),i=0..N);
```

$$telescoper := \lambda_1(n)\, u(n) + \lambda_0(n)\, u(n+1) \quad (1.1)$$

The equation we are looking for:

```
> eval(telescoper,u=unapply(a,n))=subs(k=k+1,C(k)*a)-C(k)*a;
```

$$\frac{\lambda_1(n)\binom{n}{2k}\binom{2k}{k}}{4^k} + \frac{\lambda_0(n)\binom{n+1}{2k}\binom{2k}{k}}{4^k} = \frac{C(k+1)\binom{n}{2k+2}\binom{2k+2}{k+1}}{4^{k+1}} - \frac{C(k)\binom{n}{2k}\binom{2k}{k}}{4^k} \quad (1.2)$$

Reduce it to a problem over rational functions

```
> expand(%/a);
```

$$\lambda_1(n) + \frac{\lambda_0(n)\, n}{n+1-2k} + \frac{\lambda_0(n)}{n+1-2k} = \frac{C(k+1)\, k^2}{(k+1)^2} - \frac{C(k+1)\, n\, k}{(k+1)^2} + \frac{C(k+1)\, n^2}{4\,(k+1)^2} + \frac{C(k+1)\, k}{2\,(k+1)^2} - \frac{C(k+1)\, n}{4\,(k+1)^2} - C(k) \quad (1.3)$$

reduce to the same denominator

```
> map(normal,%);
```

$$\frac{2\lambda_1(n)\, k - \lambda_1(n)\, n - \lambda_0(n)\, n - \lambda_1(n) - \lambda_0(n)}{-n-1+2k} = \frac{1}{4\,(k+1)^2}\big(4\, C(k+1)\, k^2 - 4\, C(k+1)\, n\, k + C(k+1)\, n^2 - 4\, C(k)\, k^2 + 2\, C(k+1)\, k - C(k+1)\, n - 8\, C(k)\, k - 4\, C(k)\big) \quad (1.4)$$

```
> %*denom(op(1,%))*denom(op(2,%));
```

$$4\,(k+1)^2\big(2\lambda_1(n)\, k - \lambda_1(n)\, n - \lambda_0(n)\, n - \lambda_1(n) - \lambda_0(n)\big) = (-n-1+2k)\big(4\, C(k+1)\, k^2 - 4\, C(k+1)\, n\, k + C(k+1)\, n^2 - 4\, C(k)\, k^2 + 2\, C(k+1)\, k - C(k+1)\, n - 8\, C(k)\, k - 4\, C(k)\big) \quad (1.5)$$

```
> rec:=map(collect,%,[C,lambda],factor);
```

$$rec := -4\,(k+1)^2\,(n+1)\,\lambda_0(n) + 4\,(-n-1+2k)\,(k+1)^2\,\lambda_1(n) = -4\,(-n-1+2k)\,(k+1)^2\, C(k) + (-n-1+2k)\,(2k-n+1)\,(2k-n)\, C(k+1) \quad (1.6)$$

Computation of rational solutions for the right-hand side. First, the dispersion

```
> c0:=coeff(op(2,rec),C(k));
```

$$c0 := -4\,(-n-1+2\,k)\,(k+1)^2 \qquad \textbf{(1.7)}$$

```
> c1:=subs(k=k-1,coeff(op(2,rec),C(k+1)));
```

$$c1 := (-n-3+2\,k)\,(-n-1+2\,k)\,(2\,k-2-n) \qquad \textbf{(1.8)}$$

```
> resultant(subs(k=k+h,c0),c1,k);
```

$$256\,(-4-4\,h)\,h\,(-2-4\,h)\,(-5-2\,h-n)^2\,(-3-2\,h-n)^2\,(-4-2\,h-n)^2 \qquad \textbf{(1.9)}$$

```
> roots(%,h);
```

$$\left[[-1,1],[0,1],\left[-\frac{1}{2},1\right]\right] \qquad \textbf{(1.10)}$$

Thus, 0 is the only nonnegative integer root and a multiple of the denominator is simply:

```
> den:=gcd(c0,c1);
```

$$den := -n-1+2\,k \qquad \textbf{(1.11)}$$

Recurrence for the numerator

```
> rec2:=eval(rec,C=unapply(w(k)/den,k));
```

$$rec2 := -4\,(k+1)^2\,(n+1)\,\lambda_0(n)+4\,(-n-1+2\,k)\,(k+1)^2\,\lambda_1(n) = -4\,(k+1)^2\,w(k)+(-n-1+2\,k)\,(2\,k-n)\,w(k+1) \qquad \textbf{(1.12)}$$

bound on the degree

```
> MultiSeries:-asympt(eval(op(2,rec2),w=unapply(k^r,k)),k,2)
  assuming r>0;
```

$$(-10-4\,n+4\,r)\left(\frac{1}{k}\right)^{-1-r}+\mathrm{O}\left(\left(\frac{1}{k}\right)^{-r}\right) \qquad \textbf{(1.13)}$$

So the homogeneous part does not have a polynomial solution and the upper bound on the degree of the polynomial solution has to be 2 (to reach the deg 3 of the rhs).

```
> degmax:=2:
```

Plug in undeterminate coefficients

```
> collect(eval(op(2,rec2)-op(1,rec2),w=unapply(add(c[i]*k^i,i=0.
  .degmax),k)),k);
```

$$\left((-4\,n-2)\,c_2-8\,\lambda_1(n)\right)k^3+\left(-4\,c_1-n\,(-n-1)\,c_2+(-4\,n-2)\,(2\,c_2+c_1)+4\,(n+1)\,\lambda_0(n)-4\,(-n+3)\,\lambda_1(n)\right)k^2+\left(-4\,c_1-8\,c_0-n\,(-n-1)\,(2\,c_2+c_1)+(-4\,n-2)\,(c_2+c_1+c_0)+4\,(2\,n+2)\,\lambda_0(n)+8\,\lambda_1(n)\,n\right)k-4\,c_0-n\,(-n-1)\,(c_2+c_1+c_0)+4\,(n+1)\,\lambda_0(n)-4\,(-n-1)\,\lambda_1(n) \qquad \textbf{(1.14)}$$

Extract the coefficients of k

```
> sys:=map(collect,{coeffs(%,k)},[lambda,c],factor);
```

$$sys := \{-8\,\lambda_1(n)-2\,(2\,n+1)\,c_2,\ (4\,n+4)\,\lambda_0(n)+(4\,n-12)\,\lambda_1(n)+n^2\,c_2-4\,n\,c_1-7\,n\,c_2-6\,c_1-4\,c_2,\ (4\,n+4)\,\lambda_0(n)+(4\,n+4)\,\lambda_1(n)+n^2\,c_0+n^2\,c_1+n^2\,c_2+n\,c_0+n\,c_1+n\,c_2-4\,c_0,\ (8\,n+8)\,\lambda_0(n)+8\,\lambda_1(n)\,n+n^2\,c_1+2\,n^2\,c_2-4\,n\,c_0-3\,n\,c_1-2\,n\,c_2-10\,c_0-6\,c_1-2\,c_2\} \qquad \textbf{(1.15)}$$

Solve this system that is linear in the lambda_i and the c_i.

```
> solve(sys,{seq(c[i],i=0..degmax),seq(lambda[i](n),i=0..N)});
```

$$\left\{c_0=0,\ c_1=0,\ c_2=c_2,\ \lambda_0(n)=\frac{(n+1)\,c_2}{4},\ \lambda_1(n)=-\frac{(2\,n+1)\,c_2}{4}\right\} \qquad \textbf{(1.16)}$$

And here is the telescoper that we've found:

```
> subs(%,c[2]=1,telescoper);
```

$$-\frac{(2\,n+1)\,u(n)}{4}+\frac{(n+1)\,u(n+1)}{4} \tag{1.17}$$

This is a 1st order recurrence, that is therefore easy to solve

```
> rsolve({%,u(0)=1},u(n));
```

$$\frac{2^n\,\Gamma\left(n+\frac{1}{2}\right)}{\sqrt{\pi}\;\Gamma(n+1)} \tag{1.18}$$

in other words,

```
> sol:=binomial(2*n-1,n-1)/2^(n-1);
```

$$sol := \frac{\binom{2\,n-1}{n-1}}{2^{n-1}} \tag{1.19}$$

check:

```
> normal(expand(subs(n=n+1,eval(%%%,u=unapply(sol,n)))));
```

$$0 \tag{1.20}$$